Mereology

Ben Blumson

May 21, 2018

Contents

1 Introduction

theory *Mereology*
imports *Main*
begin

This is a presentation in Isabelle/HOL of *Classical Extensional Mereology*. The presentation is based on those in "Parts" by Peter Simons [3] and "Parts and Places" by Roberto Casati and Achille Varzi [1]. Some corrections and important proofs are from [2]

Please note that this is an extremely ROUGH DRAFT.

2 Ground Mereology

Ground Mereology (M) introduces parthood as a primitive relation amongst individuals. It's assumed that parthood is a partial ordering relation - that is reflexive, symmetric and transitive [1], p. 36:

typedecl i — the type of individuals

locale M =
  fixes P:: $i \Rightarrow i \Rightarrow bool$ (**infix** $\leq$ *50*)
  assumes R: $x \leq x$ — reflexivity of parthood
    and AS: $x \leq y \longrightarrow y \leq x \longrightarrow x = y$ — antisymmetry of parthood
    and T: $x \leq y \longrightarrow y \leq z \longrightarrow x \leq z$ — transitivity of parthood

begin

The following relations are defined in terms of parthood [1], p. 36-7:

definition PP:: $i \Rightarrow i \Rightarrow bool$ (**infix** $<$ *50*)
  where $x < y \equiv x \leq y \land \neg\ y \leq x$ — proper parthood
definition O:: $i \Rightarrow i \Rightarrow bool$ (O)
  where $O\ x\ y \equiv \exists\ z.\ z \leq x \land z \leq y$ — overlap
definition D:: $i \Rightarrow i \Rightarrow bool$ (D)
  where $D\ x\ y \equiv \neg\ O\ x\ y$ — disjointness
definition U:: $i \Rightarrow i \Rightarrow bool$ (U)
  where $U\ x\ y \equiv \exists\ z.\ x \leq z \land y \leq z$ — underlap

As are the following operations on individuals [1], p. 43-5:

definition S:: $i \Rightarrow i \Rightarrow i$ (**infix** $+$ *52*)
  where $x + y \equiv THE\ z.\ \forall\ w.\ O\ w\ z \longleftrightarrow O\ w\ x \lor O\ w\ y$ — sum or fusion
definition T:: $i \Rightarrow i \Rightarrow i$ (**infix** $\times$ *53*)
  where $x \times y \equiv THE\ z.\ \forall\ w.\ O\ w\ z \longleftrightarrow O\ w\ x \land O\ w\ y$ — product or intersection

definition u:: i (u)
  where $u \equiv THE\ z.\ \forall\ w.\ P\ w\ z$ — universe
definition M:: $i \Rightarrow i \Rightarrow i$ (**infix** $-$ *51*)
  where $x - y \equiv THE\ z.\ \forall\ w.\ O\ w\ z \longleftrightarrow O\ w\ x \land \neg\ O\ w\ y$ — difference
definition C:: $i \Rightarrow i$ ($\neg$)
  where $\neg\ x \equiv (u - x)$ — complement

And the operations of general sum and product [1], p. 46:

definition σ:: $(i \Rightarrow bool) \Rightarrow i$ (σ)
  where $\sigma\ F \equiv THE\ z.\ (\forall\ y.\ O\ y\ z \longleftrightarrow (\exists\ x.\ F\ x \land O\ x\ y))$
abbreviation σx:: $(i \Rightarrow bool) \Rightarrow i$ (**binder** σ [*8*] *9*)
  where $\sigma\ x.\ F\ x \equiv \sigma\ F$ — general sum or fusion of the Fs
definition π:: $(i \Rightarrow bool) \Rightarrow i$ (π)
  where $\pi\ F \equiv THE\ z.\ (\forall\ x.\ F\ x \longrightarrow z \leq x)$ — general products [1], p. 46
abbreviation πx:: $(i \Rightarrow bool) \Rightarrow i$ (**binder** π [*8*] *9*)
  where $\pi\ x.\ F\ x \equiv \pi\ F$ — general sum or product of the Fs

Note that the symbols for part, proper part, sum, product, difference and complement are distinguished by bold font.

end

3 Minimal Mereology

Minimal mereology (MM) adds to ground mereology the axiom of weak supplementation [1], p. 39:

locale *MM* = *M* +
assumes *WS*: $x < y \longrightarrow (\exists\ z.\ z < y \wedge D\ z\ x)$ — weak supplementation

Weak supplementation is sometimes stated with parthood rather than proper parthood in the consequent. The following lemma in ground mereology shows that the two definitions are equivalent, given anti-symmetry:

lemma (**in** *M*) $(x < y \longrightarrow (\exists\ z.\ z < y \wedge D\ z\ x)) \longleftrightarrow (x < y \longrightarrow (\exists\ z.\ z \leq y \wedge D\ z\ x))$
by (*metis AS D-def O-def PP-def R*)

The following two lemmas are weaker supplementation principles taken from Simons [3], p. 27. The names *company* and *strong company* are from Varzi's *Stanford Encyclopedia of Philosophy* entry on mereology [4].

lemma (**in** *MM*) *C*: $x < y \longrightarrow (\exists\ z.\ z \neq x \wedge z < y)$ **by** (*metis WS D-def O-def R*) — company
lemma (**in** *MM*) *SC*: $x < y \longrightarrow (\exists\ z.\ z < y \wedge \neg\ z \leq x)$ **by** (*metis WS D-def O-def R*) — strong company

Minimal Mereology is not strong enough to proved the *Proper Parts Principle*, according to which if x has a proper part, and every proper part of x is a part of y, then x is a part of y [3] p. 28:

lemma (**in** *MM*) *PPP*: $\exists\ z.\ z < x \Longrightarrow \forall\ z.\ z < x \longrightarrow z \leq y \Longrightarrow x \leq y$ — proper parts principle
nitpick [*user-axioms*] **oops**

The proper parts principle is Simons way of expressing *extensionality*, which is not a theorem of Minimal Mereology either:

lemma (**in** *M*) *E*: $(\exists\ z.\ z < x \vee z < y) \longrightarrow (\forall\ z.\ z < x \longleftrightarrow z < y) \longrightarrow x = y$ — extensionality
nitpick oops

The failure of weak supplementation to entail the proper parts principle or extensionality motivates a stronger axiom, to which we turn in the next section.

4 Extensional Mereology

Extensional Mereology (EM) adds the axiom of strong supplementation [1], p. 39:

locale $EM = M +$
assumes SS: $\neg\ x \leq y \longrightarrow (\exists\ z.\ z \leq x \wedge D\ z\ y)$ — strong supplementation

Extensional Mereology (EM is so called because it entails the proper parts principle [3] p. 29:

lemma (**in** EM) PPP: $\exists\ z.\ z < x \Longrightarrow \forall\ z.\ z < x \longrightarrow z \leq y \Longrightarrow x \leq y$
by (*metis SS D-def R O-def PP-def T*)

And thus extensionality proper [1] p. 40:

lemma (**in** EM)
E: $(\exists\ z.\ z < x \vee z < y) \longrightarrow (\forall\ z.\ z < x \longleftrightarrow z < y) \longrightarrow x = y$ — extensionality

by (*metis R O-def D-def AS PP-def SS*)

In the context of the other axioms, strong supplementation entails weak supplementation [3], p. 29:

lemma (**in** M) $SStoWS$:
assumes SS: $\bigwedge x.\ \bigwedge\ y.\ \neg\ x \leq y \longrightarrow (\exists\ z.\ z \leq x \wedge D\ z\ y)$
— assumes strong supplementation
shows WS: $\bigwedge x.\ \bigwedge y.\ x < y \longrightarrow (\exists\ z.\ z < y \wedge D\ z\ x)$
— shows weak supplementation
by (*metis AS D-def O-def PP-def R assms*)

But not vice versa:

lemma (**in** M) $WStoSS$:
assumes WS: $\bigwedge x.\ \bigwedge y.\ x < y \longrightarrow (\exists\ z.\ z < y \wedge D\ z\ x)$
— assumes weak supplementation
shows SS: $\bigwedge x.\ \bigwedge\ y.\ \neg\ x \leq y \longrightarrow (\exists\ z.\ z \leq x \wedge D\ z\ y)$
— shows strong supplementation
nitpick oops

So Extensional Mereology is stronger than Minimal Mereology [1] p. 43:

sublocale $EM \subseteq MM$ **using** *T SS SStoWS* **by** (*metis MM.intro MM-axioms.intro M-axioms*)
sublocale $MM \subseteq EM$ **nitpick oops**

lemma (**in** MM)
assumes PPP: $\exists\ z.\ z < x \Longrightarrow \forall\ z.\ z < x \longrightarrow z \leq y \Longrightarrow x \leq y$
shows SS: $\neg\ x \leq y \longrightarrow (\exists\ z.\ z \leq x \wedge D\ z\ y)$ **nitpick oops**

5 Closure Mereology

Closure Mereology adds to Ground Mereology the axioms of sum closure and product closure [1] p. 43:

locale *CM* = *M* +
 assumes *SC*: $U\ x\ y \longrightarrow (\exists\ z.\ \forall\ w.\ O\ w\ z \longleftrightarrow (O\ w\ x \lor O\ w\ y))$ — sum closure

 assumes *PC*: $O\ x\ y \longrightarrow (\exists\ z.\ \forall\ w.\ w \leq z \longleftrightarrow (w \leq x \land w \leq y))$ — product closure

Combining Closure Mereology with Minimal Mereology yields the theory known as Closure Minimal Mereology *CMM* whereas combining Closure Mereology with Extensional Mereology obtains *Closed Extensional Mereology CEM* [1] p. 43:

locale *CMM* = *CM* + *MM*
locale *CEM* = *EM* + *CM*

In Closed Minimal Mereology, the product closure axiom and weak supplementation jointly entail strong supplementation. The proof verified here is from Pontow [2] p. 200:

lemma (**in** *CMM*) *SS*: $\neg\ x \leq y \longrightarrow (\exists\ z.\ z \leq x \land D\ z\ y)$
proof fix $x\ y$
 assume $\neg\ x \leq y$
 show $(\exists\ z.\ z \leq x \land D\ z\ y)$
 proof *cases*
  assume $D\ x\ y$
  thus $(\exists\ z.\ z \leq x \land D\ z\ y)$ **using** *R* **by** *auto*
 next
  assume $\neg\ D\ x\ y$
  hence $O\ x\ y$ **using** *D-def* **by** *simp*
  hence $\exists\ z.\ \forall\ w.\ w \leq z \longleftrightarrow (w \leq x \land w \leq y)$ **using** *PC* **by** *simp*
  then obtain z **where** z: $\forall\ w.\ w \leq z \longleftrightarrow (w \leq x \land w \leq y)$**..**
  hence $z < x$ **using** *PP-def R* $\langle\neg\ x \leq y\rangle$ **by** *auto*
  hence $(\exists\ w.\ w < x \land D\ w\ z)$ **using** *WS* **by** *simp*
  then obtain w **where** $w < x \land D\ w\ z$**..**
  hence $w \leq x \land D\ w\ y$ **by** (*meson D-def O-def PP-def T z*)
  thus $(\exists\ z.\ z \leq x \land D\ z\ y)$**..**
 qed
qed

Because Strong Supplementation is provable in Closed Minimal Mereology, it follows that Closed Extensional Mereology and Closed Minimal Mereology are the same theory [1] p. 44:

sublocale *CEM* ⊆ *CMM* **by** (*simp add*: *CMM.intro CM-axioms MM-axioms*)
sublocale *CMM* ⊆ *CEM* **by** (*simp add*: *CEM.intro CM-axioms EM.intro EM-axioms.intro M-axioms SS*)

Closure Mereology with Universe (CMU) is obtained by adding an axiom ensuring existence of a universe [1] p. 44:

locale *CMU* = *CM* +
  assumes *U*: $\exists\ z.\ \forall\ x.\ x \leq z$ — universe

And adding Extensional Mereology (or Minimal Mereology) to this theory results in Closed Extensional Mereology with Universe *CEMU*:

locale *CEMU* = *EM* + *CMU*

In Closure Extensional Mereology with Universe, it's possible to derive a strengthening of the sum axiom, since everything underlaps everything else:

lemma (**in** *CEMU*) *EU*: $U\ x\ y$ **using** *U-def U* **by** *auto* — everything underlaps
lemma (**in** *CEMU*) *SSC*: $(\exists\ z.\ \forall\ w.\ O\ w\ z \longleftrightarrow (O\ w\ x \lor O\ w\ y))$ **using** *EU SC* **by** *simp*
 — strengthened sum closure

6 General Mereology

General Mereology is obtained from Ground Mereology by adding the axiom of fusion or unrestricted composition [1] p. 46:

locale *GM* = *M* +
  assumes *F*: $(\exists\ x.\ F\ x) \longrightarrow (\exists\ z.\ \forall\ y.\ O\ y\ z \longleftrightarrow (\exists\ x.\ F\ x \land O\ x\ y))$ — fusion or unrestricted composition

Substituting $x = a \lor x = b$ for $F\ x$ in the fusion axiom allows the derivation of an unrestricted version of sum closure *GS*, and so of course sum closure itself, as follows:

lemma (**in** *GM*) *FS*:
$(\exists\ x.\ (x = a \lor x = b)) \longrightarrow (\exists\ z.\ \forall\ y.\ O\ y\ z \longleftrightarrow (\exists\ x.\ (x = a \lor x = b) \land O\ x\ y))$
  using *F* **solve-direct.**
lemma (**in** *GM*) *GFS*: $(\exists\ z.\ \forall\ y.\ O\ y\ z \longleftrightarrow (\exists\ x.\ (x = a \lor x = b) \land O\ x\ y))$
  using *FS* **by** *blast*
lemma (**in** *M*) *GFStoGS*:
  assumes *GFS*: $(\exists\ z.\ \forall\ y.\ O\ y\ z \longleftrightarrow (\exists\ x.\ (x = a \lor x = b) \land O\ x\ y))$
  shows $(\exists z.\ \forall w.\ O\ w\ z \longleftrightarrow (O\ w\ a \lor O\ w\ b))$ **by** (*metis O-def GFS*)
lemma (**in** *GM*) *GS*: $(\exists z.\ \forall w.\ O\ w\ z \longleftrightarrow (O\ w\ x \lor O\ w\ y))$
  using *GFS GFStoGS* **by** *simp*
lemma (**in** *GM*) *S*: $U\ x\ y \longrightarrow (\exists z.\ \forall w.\ O\ w\ z \longleftrightarrow (O\ w\ x \lor O\ w\ y))$
  using *GS* **by** *simp*

But product closure cannot be derived:

lemma (**in** *GM*) *T*: $O\ x\ y \longrightarrow (\exists\ z.\ \forall\ w.\ w \leq z \longleftrightarrow (w \leq x \land w \leq y))$
  nitpick [*show-all*] **oops**

It follows that General Mereology does not encompass Closure Mereology, contrary to Simons [3] p. 36 and Casati and Varzi [1] p. 46 (this point is discussed in detail by Pontow [2]:

sublocale $GM \subseteq CM$ **nitpick** [*show-all*] **oops**

It's possible to prove from fusion in General Mereology that there is something that overlaps everything:

lemma (**in** GM) $\exists\ z.\ \forall\ x.\ O\ x\ z$ — something overlaps everything
proof –
  have $(\exists\ x.\ x = x) \longrightarrow (\exists\ z.\ \forall\ y.\ O\ y\ z \longleftrightarrow (\exists\ x.\ x = x \wedge O\ x\ y))$ **using** F **by** *fast*
  hence $\exists\ z.\ \forall\ y.\ O\ y\ z \longleftrightarrow (\exists\ x.\ x = x \wedge O\ x\ y)$ **by** *simp*
  hence $\exists\ z.\ \forall\ y.\ O\ y\ z \longleftrightarrow (\exists\ x.\ O\ x\ y)$ **by** *simp*
  thus *?thesis* **by** (*metis O-def R*)
qed

But it doesn't follow that there is a universe, let alone a unique universe. If for example, there is just an infinite ascending chain, then everything overlaps everything else, but there isn't a particular thing which everything is a part of, since for anything in particular, the things above it are not part of the chain:

lemma (**in** GM) U: $\exists\ z.\ \forall\ x.\ x \leq z$ **nitpick oops**

The existence of differences is not guaranteed either:

lemma (**in** GM) D: $(\exists\ w.\ w \leq x \wedge \neg\ O\ w\ y) \longrightarrow (\exists\ z.\ \forall\ w.\ w \leq z \longleftrightarrow (w \leq x \wedge \neg\ O\ w\ y))$ **nitpick oops**

Call the combination of General Mereology with weak supplementation General Minimal Mereology, or GMM:

locale $GMM = MM + GM$ — General Minimal Mereology

Although Strong Supplementation can be derived from Weak Supplementation in Closed Minimal Mereology via the product axioms, it cannot be derived in General Minimal Mereology, since the product axiom itself still cannot be derived in General Minimal Mereology:

lemma (**in** GMM) SS: $\neg\ x \leq y \longrightarrow (\exists\ z.\ z \leq x \wedge D\ z\ y)$ **nitpick oops**
lemma (**in** GMM) T: $O\ x\ y \longrightarrow (\exists\ z.\ \forall\ w.\ w \leq z \longleftrightarrow (w \leq x \wedge w \leq y))$ **nitpick** [*show-all*] **oops**

Nor can the existence of a universe or differences be proved in General Minimal Mereology:

lemma (**in** GMM) U: $\exists\ z.\ \forall\ x.\ x \leq z$ **nitpick oops**
lemma (**in** GMM) D: $(\exists\ w.\ w \leq x \wedge \neg\ O\ w\ y) \longrightarrow (\exists\ z.\ \forall\ w.\ w \leq z \longleftrightarrow (w \leq x \wedge \neg\ O\ w\ y))$ **nitpick oops**

7 Classical Extensional Mereology

Classical Extensional Mereology *GEM* is simply Extensional Mereology combined with General Mereology [1] p. 46:

locale $GEM = EM + GM$

The presence of strong supplementation in Classical Extensional Mereology enables the derivation of product closure from fusion. The following proof is from Pontow [2] pp. 202-3.

The proof begins by substitutions $z \leq a \wedge z \leq b$ for F in the fusion axiom, to give the existence of a sum of all the parts of a and b:

lemma (**in** *GM*) *FP*: $(\exists\ z.\ (z \leq a \wedge z \leq b)) \longrightarrow (\exists\ z.\ \forall\ w.\ O\ w\ z \longleftrightarrow (\exists\ x.\ (x \leq a \wedge x \leq b) \wedge O\ x\ w))$
using F **solve-direct.**

Three lemmas are helpful before proceeding with the proof proper. First, strong supplementation is needed to proceed from the fact that z is *a* sum of the Fs to the fact that z is *the* sum of the Fs:

lemma (**in** *EM*) *atothesum*:
assumes *asum*: $\forall\ y.\ O\ y\ z \longleftrightarrow (\exists\ x.\ F\ x \wedge O\ x\ y)$
shows *thesum*: $(THE\ v.\ \forall\ y.\ O\ y\ v \longleftrightarrow (\exists\ x.\ F\ x \wedge O\ x\ y)) = z$
proof (*rule the-equality*)
show $\forall\ y.\ O\ y\ z \longleftrightarrow (\exists\ x.\ F\ x \wedge O\ x\ y)$ **using** *asum*.
show $\bigwedge v.\ \forall y.\ O\ y\ v = (\exists x.\ F\ x \wedge O\ x\ y) \Longrightarrow v = z$ **by** (*metis SS AS D-def O-def R asum*)
qed

Using this lemma, we can show that if something overlaps z just in case it overlaps an F, then it is the sum of the Fs:

lemma (**in** *EM*) *UGS*: $(\forall\ y.\ O\ y\ z \longleftrightarrow (\exists\ x.\ F\ x \wedge O\ x\ y)) \longrightarrow (\sigma\ x.\ F\ x) = z$
proof
assume $(\forall\ y.\ O\ y\ z \longleftrightarrow (\exists\ x.\ F\ x \wedge O\ x\ y))$
hence $(THE\ v.\ \forall\ y.\ O\ y\ v \longleftrightarrow (\exists\ x.\ F\ x \wedge O\ x\ y)) = z$ **using** *atothesum* **by** *simp*
thus $(\sigma\ v.\ F\ v) = z$ **using** σ-*def* **by** *blast*
qed

With this lemma in hand, we can proceed with a final lemma the proof from Pontow [2] pp. 202-3, according to which if there is an F, then everything is part of the sum of the Fs just in case every part of it overlaps with an F.

lemma (**in** *GEM*) *PS*: $(\exists\ x.\ F\ x) \longrightarrow (\forall\ y.\ y \leq (\sigma\ v.\ F\ v) \longleftrightarrow (\forall\ w.\ w \leq y \longrightarrow (\exists\ v.\ F\ v \wedge O\ v\ w)))$
proof
assume $(\exists\ x.\ F\ x)$
hence $\exists\ z.\ \forall\ y.\ O\ y\ z \longleftrightarrow (\exists\ x.\ F\ x \wedge O\ x\ y)$ **using** F **by** *simp*
then obtain z **where** z: $\forall\ y.\ O\ y\ z \longleftrightarrow (\exists\ x.\ F\ x \wedge O\ x\ y)$**..**

```
  hence σ: (σ v. F v) = z  using UGS by simp
  show ∀ y. y ≤ (σ v. F v) ⟷ (∀ w. w ≤ y ⟶ (∃ v. F v ∧ O v w))
  proof
    fix y
    show y ≤ (σ v. F v) ⟷ (∀ w. w ≤ y ⟶ (∃ v. F v ∧ O v w))
      proof
        assume y ≤ (σ v. F v)
        hence y ≤ z using σ by simp
        hence O y z using O-def R by auto
        hence (∃ x. F x ∧ O x y) using z by simp
       thus (∀ w. w ≤ y ⟶ (∃ v. F v ∧ O v w)) by (metis O-def R T ‹y ≤ z› z)
      next
        assume (∀ w. w ≤ y ⟶ (∃ v. F v ∧ O v w))
        hence y ≤ z using z by (meson D-def SS)
        thus y ≤ (σ v. F v) using σ by simp
      qed
    qed
  qed
```

Continuing to follow the proof from [2] pp. 204, we can prove the Product Axiom proper:

```
lemma (in GEM) T: O x y ⟶ (∃ z. ∀ w. w ≤ z ⟷ (w ≤ x ∧ w ≤ y))
proof
  assume O x y
  hence ez:  (∃ z. (z ≤ x ∧ z ≤ y)) using O-def by simp
  hence (∃ z. ∀ w. O w z ⟷ (∃ v. (v ≤ x ∧ v ≤ y) ∧ O v w)) using FP by
simp
  then obtain z where (∀ w. O w z ⟷ (∃ v. (v ≤ x ∧ v ≤ y) ∧ O v w))..
  hence σzxy: (σ v. v ≤ x ∧ v ≤ y) = z using UGS by simp
  have gragra: (∀ s. s ≤ (σ v. v ≤ x ∧ v ≤ y) ⟷
(∀ w. w ≤ s ⟶ (∃ v. v ≤ x ∧ v ≤ y ∧ O v w))) using PS ez by simp
  have ∀ w. w ≤ z ⟷ (w ≤ x ∧ w ≤ y)
  proof
    fix w
    show w ≤ z ⟷ (w ≤ x ∧ w ≤ y)
      proof
        assume w ≤ z
        hence w ≤ (σ v. v ≤ x ∧ v ≤ y) using σzxy by simp
         hence dadada: (∀ t. t ≤ w ⟶ (∃ v. v ≤ x ∧ v ≤ y ∧ O v t)) using
gragra by simp
        have ∀ t. t ≤ w ⟶ (O t x ∧ O t y)
        proof
          fix t
          show t ≤ w ⟶ (O t x ∧ O t y)
            proof
              assume t ≤ w
              hence (∃ v. v ≤ x ∧ v ≤ y ∧ O v t) using dadada by simp
              thus O t x ∧ O t y using O-def T by blast
            qed
```

    qed
    thus $w \leq x \wedge w \leq y$ **using** *SS T D-def* **by** *meson*
  next
    assume $w \leq x \wedge w \leq y$
    thus $w \leq z$ **using** *O-def R σzxy gragra* **by** *fastforce*
  qed
 qed
 thus $(\exists\ z.\ \forall\ w.\ w \leq z \longleftrightarrow (w \leq x \wedge w \leq y))$**..**
qed

It follows that General Extensional Mereology is stronger than Closed Extensional Mereology

sublocale *GEM* ⊆ *CEM* **using** *CEM.intro CM.intro CM-axioms-def EM-axioms M-axioms S T* **by** *blast*

Likewise, substituting $x = x$ for $F\ x$ in fusion allows the derivation of the existence of a universe:

lemma (**in** *GM*) *selfidentity*:
$(\exists\ x.\ x = x) \longrightarrow (\exists\ z.\ \forall\ y.\ O\ y\ z \longleftrightarrow (\exists\ x.\ x = x \wedge O\ x\ y))$
 using *F* **by** *fast*
lemma (**in** *GM*) $(\exists\ z.\ \forall\ y.\ O\ y\ z \longleftrightarrow (\exists\ x.\ O\ x\ y))$ **using** *selfidentity* **by** *simp*
lemma (**in** *GEM*) *U*: $\exists\ z.\ \forall\ x.\ x \leq z$
 using *selfidentity* **by** (*metis D-def O-def SS*)

It follows that Classical Extensional Mereology is also stronger than Closed Extensional Mereology with Universe:

sublocale *GEM* ⊆ *CEMU*
proof
 show $\exists z.\ \forall x.\ x \leq z$ **using** *U* **by** *simp*
qed

The existence of differences is also derivable in General Extensional Mereology. Like, the proof of the product axiom, the proof of the existence of differences is quite involved. It can be found in Pontow [2] p. 209.

lemma (**in** *GM*) *FD*: $(\exists\ x.\ x \leq a \wedge \neg\ O\ x\ b) \longrightarrow (\exists\ z.\ \forall\ y.\ O\ y\ z \longleftrightarrow (\exists\ x.\ (x \leq a \wedge \neg\ O\ x\ b) \wedge O\ x\ y))$ **using** *F* **solve-direct.**

lemma (**in** *GEM*) *D*: $(\exists\ w.\ w \leq x \wedge \neg\ O\ w\ y) \longrightarrow (\exists\ z.\ \forall\ w.\ w \leq z \longleftrightarrow (w \leq x \wedge \neg\ O\ w\ y))$
proof
 assume $(\exists\ w.\ w \leq x \wedge \neg\ O\ w\ y)$
 hence $(\exists\ z.\ \forall\ w.\ O\ w\ z \longleftrightarrow (\exists\ v.\ (v \leq x \wedge \neg\ O\ v\ y) \wedge O\ v\ w))$ **using** *FD* **by** *simp*
 then obtain Σ **where** Σ: $\forall\ w.\ O\ w\ \Sigma \longleftrightarrow (\exists\ v.\ (v \leq x \wedge \neg\ O\ v\ y) \wedge O\ v\ w)$**..**
 have $\forall\ w.\ w \leq \Sigma \longleftrightarrow (w \leq x \wedge \neg\ O\ w\ y)$
 proof
  fix *w*
  show $w \leq \Sigma \longleftrightarrow (w \leq x \wedge \neg\ O\ w\ y)$

proof
assume $w \leq \Sigma$
have $\forall\ z.\ z \leq w \longrightarrow O\ z\ x$
proof
fix z
show $z \leq w \longrightarrow O\ z\ x$
proof
assume $z \leq w$
hence $\exists\ s0.\ (s0 \leq x \wedge \neg\ O\ s0\ y) \wedge O\ s0\ z$ **using** *M.T M-axioms O-def R* Σ ‹$w \leq \Sigma$› **by** *blast*
thus $O\ z\ x$ **using** *M.T M-axioms O-def* **by** *blast*
qed
qed
hence $w \leq x$ **using** *SS D-def* **by** *blast*
have $\forall\ v.\ v \leq w \longrightarrow \neg\ v \leq y$ **by** (*metis O-def M.T M-axioms* Σ ‹$w \leq \Sigma$›)
hence $\neg\ O\ w\ y$ **using** *O-def* **by** *simp*
thus $w \leq x \wedge \neg\ O\ w\ y$ **using** ‹$w \leq x$› **by** *blast*
next
assume $w \leq x \wedge \neg\ O\ w\ y$
show $w \leq \Sigma$ **by** (*meson D-def O-def R SS* Σ ‹$w \leq x \wedge \neg\ O\ w\ y$›)
qed
qed
thus $(\exists\ z.\ \forall\ w.\ w \leq z \longleftrightarrow (w \leq x \wedge \neg\ O\ w\ y))$..
qed

8 Atomism

An atom is an individual with no proper parts:

definition (**in** *M*) *A*:: *i* $\Rightarrow$ *bool* (*A*)
where $A\ x\ \equiv \neg\ (\exists\ y.\ y < x)$

Each theory discussed above can be augmented with an axiom stating that everything has an atom as a part, viz. [1], p. 48:

locale *AM* = *M* +
assumes *A*: $\forall\ x.\ \exists\ y.\ A\ y \wedge y \leq x$ — atomicity
locale *AMM* = *AM* + *MM*
locale *AEM* = *AM* + *EM*
locale *ACM* = *AM* + *CM*
locale *ACEM* = *AM* + *CEM*
locale *AGM* = *AM* + *GM*
locale *AGEM* = *AM* + *GEM*

It follows in *AEM* that if something is not part of another, there is an atom which is part of the former but not part of the later:

lemma (**in** *AEM*)
ASS: $\neg\ x \leq y \longrightarrow (\exists\ z.\ A\ z \wedge (z \leq x \wedge \neg\ O\ z\ y))$

proof
  assume $\neg\ x \leq y$
  hence $(\exists\ w.\ w \leq x \wedge D\ w\ y)$ **using** *SS* **by** *simp*
  then obtain w **where** w: $w \leq x \wedge D\ w\ y$**..**
  hence $\exists\ z.\ A\ z \wedge z \leq w$ **using** *A* **by** *simp*
  then obtain z **where** z: $A\ z \wedge z \leq w$**..**
  hence $A\ z \wedge (z \leq x \wedge \neg\ O\ z\ y)$ **by** (*meson D-def O-def T w*)
  thus $\exists\ z.\ A\ z \wedge (z \leq x \wedge \neg\ O\ z\ y)$**..**
qed

Moreover, in Minimal Mereology this lemma entails both strong supplementation and atomism, so it serves as an alternative characterisation of Atomistic Extensional Mereology:

lemma (**in** *MM*)
  assumes *ASS*: $\neg\ x \leq y \longrightarrow (\exists\ z.\ A\ z \wedge (z \leq x \wedge \neg\ O\ z\ y))$
  shows *SS*: $\neg\ x \leq y \longrightarrow (\exists\ z.\ z \leq x \wedge D\ z\ y)$ **using** *D-def assms* **by** *blast*

lemma (**in** *M*)
  assumes *ASS*: $\forall\ x.\ \forall\ y.\ \neg\ x \leq y \longrightarrow (\exists\ z.\ A\ z \wedge (z \leq x \wedge \neg\ O\ z\ y))$
  shows *A*: $\forall\ x.\ \exists\ y.\ A\ y \wedge y \leq x$
proof
  fix x
  show $\exists\ y.\ A\ y \wedge y \leq x$
  proof *cases*
    assume $A\ x$
    hence $A\ x \wedge x \leq x$ **using** *R* **by** *simp*
    thus $\exists\ y.\ A\ y \wedge y \leq x$**..**
  next
    assume $\neg\ A\ x$
    hence $\exists\ y.\ y < x$ **using** *A-def* **by** *simp*
    then obtain y **where** y: $y < x$**..**
    hence $\neg\ x \leq y$ **using** *PP-def* **by** *simp*
    hence $\exists\ z.\ A\ z \wedge (z \leq x \wedge \neg\ O\ z\ y)$ **using** *ASS* **by** *blast*
    thus $\exists\ y.\ A\ y \wedge y \leq x$ **by** *blast*
  qed
qed

So the axiom of Atomistic Strong Supplementation could be used in place of the two axioms of Atomicity and Strong Supplementation [1]

For the same reason that the Product axiom and Strong Supplementation do not follow from the Fusion Axiom in General Mereology, and so General Mereology is strictly weaker than Classical Extensional Mereology, the Product Axiom and Strong Supplementation still do not follow from the Fusion Axiom in Atomistic General Mereology, and so Atomistic General Mereology is also strictly weaker than Atomistic Classical Extensional Mereology [2], p. 206:

lemma (**in** *AGM*) *T*: $O\ x\ y \longrightarrow (\exists\ z.\ \forall\ w.\ w \leq z \longleftrightarrow (w \leq x \wedge w \leq y))$ **nitpick**

[*user-axioms*] **oops**
lemma (**in** *AGM*) *SS*: $\neg\ x \leq y \longrightarrow (\exists\ z.\ z \leq x \land D\ z\ y)$ **nitpick** [*user-axioms*]
oops

Alternatively, each theory discussed above can be augmented with an axiom stating that there are no atoms, viz.:

locale *XAM* = *M* +
  assumes *XA*: $\neg\ (\exists\ x.\ A\ x)$ — atomlessness
locale *XAMM* = *XAM* + *MM*
locale *XAEM* = *XAM* + *EM*
locale *XACM* = *XAM* + *CM*
locale *XACEM* = *XAM* + *CEM*
locale *XAGM* = *XAM* + *GM*
locale *XAGEM* = *XAM* + *GEM*

Pontow notes that the question of whether the Fusion Axiom entails the Product and Strong Supplementation axioms in Atomless General Mereology is open [2]. Nitpick does not find a countermodel (since an infinite countermodel is needed?) and sledgehammer fails to find a proof, so this problem remains open for now:

lemma (**in** *XAGM*) *T*: $O\ x\ y \longrightarrow (\exists\ z.\ \forall\ w.\ w \leq z \longleftrightarrow (w \leq x \land w \leq y))$ **oops**
lemma (**in** *XAGM*) *SS*: $\neg\ x \leq y \longrightarrow (\exists\ z.\ z \leq x \land D\ z\ y)$ **oops**

9 Consistency

I conclude by proving the consistency of all the theories mentioned.

lemma (**in** *M*) *False* **nitpick** [*show-all*] **oops**
lemma (**in** *MM*) *False* **nitpick** [*show-all*] **oops**
lemma (**in** *EM*) *False* **nitpick** [*show-all*] **oops**
lemma (**in** *CM*) *False* **nitpick** [*show-all*] **oops**
lemma (**in** *CEM*) *False* **nitpick** [*show-all*] **oops**
lemma (**in** *CMU*) *False* **nitpick** [*show-all*] **oops**
lemma (**in** *CEMU*) *False* **nitpick** [*show-all*] **oops**
lemma (**in** *GM*) *False* **nitpick** [*show-all*] **oops**
lemma (**in** *GEM*) *False* **nitpick** [*show-all*] **oops**

end

References

[1] R. Casati and A. C. Varzi. *Parts and Places: The Structures of Spatial Representation.* MIT Press, Cambridge, Mass., 1999.

[2] C. Pontow. A note on the axiomatics of theories in parthood. *Data & Knowledge Engineering*, 50(2):195–213, Aug. 2004.

[3] P. Simons. *Parts: A Study in Ontology.* Clarendon Press, Oxford, 1987.

[4] A. Varzi. Mereology. In E. N. Zalta, editor, *The Stanford Encyclopedia of Philosophy.* Metaphysics Research Lab, Stanford University, winter 2016 edition, 2016.