Gödel's God in Isabelle/HOL

Christoph Benzmüller and Bruno Woltzenlogel Paleo

November 9, 2013

A1	Either a property or its negation is positive, but not both:	$\forall\phi[P(\neg\phi) \leftrightarrow \neg P(\phi)]$
A2	A property necessarily implied by a positive property is positive:	$\forall\phi\forall\psi[(P(\phi) \wedge \Box\forall x[\phi(x) \rightarrow \psi(x)]) \rightarrow P(\psi)]$
T1	Positive properties are possibly exemplified:	$\forall\varphi[P(\varphi) \rightarrow \Diamond\exists x\varphi(x)]$
D1	A *God-like* being possesses all positive properties:	$G(x) \leftrightarrow \forall\phi[P(\phi) \rightarrow \phi(x)]$
A3	The property of being God-like is positive:	$P(G)$
C	Possibly, God exists:	$\Diamond\exists x G(x)$
A4	Positive properties are necessarily positive:	$\forall\phi[P(\phi) \rightarrow \Box\, P(\phi)]$
D2	An *essence* of an individual is a property possessed by it and necessarily implying any of its properties:	$\phi\ ess.\ x \leftrightarrow \phi(x) \wedge \forall\psi(\psi(x) \rightarrow \Box\forall y(\phi(y) \rightarrow \psi(y)))$
T2	Being God-like is an essence of any God-like being:	$\forall x[G(x) \rightarrow G\ ess.\ x]$
D3	*Necessary existence* of an individual is the necessary exemplification of all its essences:	$NE(x) \leftrightarrow \forall\phi[\phi\ ess.\ x \rightarrow \Box\exists y\phi(y)]$
A5	Necessary existence is a positive property:	$P(NE)$
T3	Necessarily, God exists:	$\Box\exists x G(x)$

1 Introduction

Dana Scott's version [11] of Goedel's ontological argument [8] for God's existence is here formalized in quantified modal logic KB (QML KB) within the proof assistant Isabelle/HOL. QML KB is modeled as a fragment of classical higher-order logic (HOL); thus, the formalization is essentially a formalization in HOL. The employed embedding of QML KB in HOL is adapting the work of Benzmüller and Paulson [2, 1]. Note that the QML KB formalization employs quantification over individuals and quantification over sets of individuals (properties).

The gaps in Scott's proof have been automated with Sledgehammer [5], performing remote calls to the higher-order automated theorem prover LEO-II [3]. Sledgehammer then suggests the Metis [9] calls. The Metis proofs are verified by Isabelle/HOL. For consistency checking, the model finder Nitpick [6] has been employed.

Isabelle is described in the textbook by Nipkow, Paulson, and Wenzel [10] and in tutorials available at: http://isabelle.in.tum.de.

1.1 Related Work

The formalization presented here is related to the THF [13] and Coq [4] formalizations at https://github.com/FormalTheology/GoedelGod/tree/master/Formalizations/.
A medieval ontological argument by Anselm was formalized in PVS by John Rushby [?].

2 An Embedding of QML KB in HOL

The types i for possible worlds and μ for individuals are introduced.

typedecl i — the type for possible worlds
typedecl μ — the type for indiviuals

Possible worlds are connected by an accessibility relation r.

consts r :: i ⇒ i ⇒ bool (**infixr** r 70) — accessibility relation r

The B axiom (symmetry) for relation r is stated. B is needed only for proving theorem T3.

axiomatization where sym: x r y ⟶ y r x

QML formulas are translated as HOL terms of type $i \Rightarrow$ *bool*. This type is abbreviated as σ.

type-synonym σ = (i ⇒ bool)

The classical connectives $\neg, \wedge, \rightarrow$, and $\forall$ (over individuals and over sets of individuals) and $\exists$ (over individuals) are lifted to type σ. The lifted connectives are $m\neg$, $m\wedge$, $m\rightarrow$, $\forall$, and $\exists$ (the latter two are modeled as constant symbols). Other connectives could be introduced analogously. Definitions could be used instead of abbreviations.

abbreviation mnot :: $\sigma \Rightarrow \sigma$ (m¬) **where** m¬ $\varphi \equiv (\lambda$w. ¬ φ w)
abbreviation mand :: $\sigma \Rightarrow \sigma \Rightarrow \sigma$ (**infixr** m∧ 79) **where** φ m∧ $\psi \equiv (\lambda$w. φ w ∧ ψ w)
abbreviation mimplies :: $\sigma \Rightarrow \sigma \Rightarrow \sigma$ (**infixr** m→ 74) **where** φ m→ $\psi \equiv (\lambda$w. φ w ⟶ ψ w)
abbreviation mforall :: ('a ⇒ σ) ⇒ σ (∀) **where** ∀ Φ ≡ (λw. ∀x. Φ x w)
abbreviation mexists :: ('a ⇒ σ) ⇒ σ (∃) **where** ∃ Φ ≡ (λw. ∃x. Φ x w)
abbreviation mbox :: $\sigma \Rightarrow \sigma$ (□) **where** □ $\varphi \equiv (\lambda$w. ∀v. w r v ⟶ φ v)
abbreviation mdia :: $\sigma \Rightarrow \sigma$ (◇) **where** ◇ $\varphi \equiv (\lambda$w. ∃v. w r v ∧ φ v)

For grounding lifted formulas, the meta-predicate *valid* is introduced.

abbreviation valid :: σ ⇒ bool ([-]) **where** [p] ≡ ∀w. p w

3 Gödel's Ontological Argument

Constant symbol P (Gödel's 'Positive') is declared.

consts P :: ($\mu \Rightarrow \sigma$) ⇒ σ

The meaning of P is restricted by axioms *A1*(a/b): $\forall\varphi[P(\neg\varphi) \leftrightarrow \neg P(\varphi)]$ (Either a property or its negation is positive, but not both.) and *A2*: $\forall\varphi\forall\psi[(P(\varphi)\wedge\Box\forall x[\varphi(x) \rightarrow \psi(x)]) \rightarrow P(\psi)]$ (A property necessarily implied by a positive property is positive).

axiomatization where

A1a: [$\forall$ ($\lambda\varphi$. P (λx. m$\neg$ (φ x)) m$\rightarrow$ m$\neg$ (P φ))] **and**

A1b: [$\forall$ ($\lambda\varphi$. m$\neg$ (P φ) m$\rightarrow$ P (λx. m$\neg$ (φ x)))] **and**

A2: [$\forall$ ($\lambda\varphi$. $\forall$ ($\lambda\psi$. (P φ m$\wedge$ $\Box$ ($\forall$ (λx. φ x m$\rightarrow$ ψ x))) m$\rightarrow$ P ψ))]

We prove theorem T1: $\forall\varphi[P(\varphi) \rightarrow \Diamond\exists x\varphi(x)]$ (Positive properties are possibly exemplified). T1 is proved directly by Sledgehammer with command *sledgehammer* [*provers* = *remote-leo2*]. Sledgehammer suggests to call Metis with axioms A1a and A2. Metis sucesfully generates a proof object that is verified in Isabelle/HOL's kernel.

theorem T1: [$\forall$ ($\lambda\varphi$. P φ m$\rightarrow$ $\Diamond$ ($\exists$ φ))]
sledgehammer [provers = remote-leo2]
by (metis A1a A2)

Next, the symbol G for 'God-like' is introduced and defined as $G(x) \leftrightarrow \forall\varphi[P(\phi) \rightarrow \varphi(x)]$ (A God-like being possesses all positive properties).

definition G :: $\mu \Rightarrow \sigma$ **where** G = (λx. $\forall$ ($\lambda\varphi$. P φ m$\rightarrow$ φ x))

Axiom *A3* is added: $P(G)$ (The property of being God-like is positive). Sledgehammer and Metis then prove corollary C: $\Diamond\exists xG(x)$ (Possibly, God exists).

axiomatization where A3: [P G]

corollary C: [$\Diamond$ ($\exists$ G)]
sledgehammer [provers = remote-leo2] **by** (metis A3 T1)

Axiom *A4* is added: $\forall\phi[P(\phi) \rightarrow \Box\, P(\phi)]$ (Positive properties are necessarily positive).

axiomatization where A4: [$\forall$ ($\lambda\varphi$. P φ m$\rightarrow$ $\Box$ (P φ))]

Symbol *ess* for 'Essence' is introduced and defined as $\varphi\ ess.\ x \leftrightarrow \varphi(x) \wedge \forall\psi(\psi(x) \rightarrow \Box\forall y(\varphi(y) \rightarrow \psi(y)))$ (An *essence* of an individual is a property possessed by it and necessarily implying any of its properties).

definition ess :: ($\mu \Rightarrow \sigma$) $\Rightarrow \mu \Rightarrow \sigma$ (**infixr** ess 85) **where**
φ ess x = φ x m$\wedge$ $\forall$ ($\lambda\psi$. ψ x m$\rightarrow$ $\Box$ ($\forall$ (λy. φ y m$\rightarrow$ ψ y)))

Next, Sledgehammer and Metis prove theorem *T2*: $\forall x[G(x) \rightarrow G\ ess.\ x]$ (Being God-like is an essence of any God-like being).

theorem T2: [$\forall$ (λx. G x m$\rightarrow$ G ess x)]
sledgehammer [provers = remote-leo2] **by** (metis A1b A4 G-def ess-def)

Symbol *NE*, for 'Necessary Existence', is introduced and defined as $NE(x) \leftrightarrow \forall\varphi[\varphi\ ess.\ x \rightarrow \Box\exists y\varphi(y)]$ (Necessary existence of an individual is the necessary exemplification of all its essences).

definition NE :: $\mu \Rightarrow \sigma$ **where** NE = (λx. $\forall$ ($\lambda\varphi$. φ ess x m$\rightarrow$ $\Box$ ($\exists$ φ)))

Moreover, axiom *A5* is added: $P(NE)$ (Necessary existence is a positive property).

axiomatization where A5: [P NE]

Finally, Sledgehammer and Metis prove the main theorem *T3*: $\Box\exists x G(x)$ (Necessarily, God exists).

theorem T3: [□ (∃ G)]
sledgehammer [provers = remote-leo2] **by** (metis A5 C T2 sym G-def NE-def)

corollary C2: [∃ G]
sledgehammer [provers = remote-leo2](T1 T3 G-def sym) **by** (metis T1 T3 G-def sym)

The consistency of the entire theory is checked with Nitpick.

lemma True **nitpick** [satisfy, user-axioms, expect = genuine] **oops**

4 Further results on Gödel's God.

Lifted Leibniz equality is introduced.

abbreviation *mequals* :: $\mu \Rightarrow \mu \Rightarrow \sigma$ (**infixr** *m=* *90*) **where** x *m=* $y \equiv \forall(\lambda\ \varphi.(\varphi\ x\ m{\rightarrow}\ \varphi\ y))$

Gödel's God is flawless, that is, he has only positive properties.

theorem *Flawless*: $[\forall(\lambda\varphi.\ \forall(\lambda x.\ (G\ x\ m{\rightarrow}\ (m\neg\ (P\ \varphi)\ m{\rightarrow}\ m\neg\ (\varphi\ x)))))]$
sledgehammer [*provers = remote-leo2 remote-satallax*] **by** (*metis A1b G-def*)

Moreover, it can be shown that any two God-like beings are equal, that is, there is only one God-like being.

theorem *Monotheism*: $[\forall(\lambda x.\ \forall(\lambda y.\ (G(x)\ m{\rightarrow}\ (G(y)\ m{\rightarrow}\ (x\ m{=}\ y)))))]$
sledgehammer [*provers = remote-leo2*] **by** (*metis Flawless G-def*)

Add-on: We briefly show that lifted Leibniz equality indeed denotes equality.

lemma *eqtest1*: $x = y \Longrightarrow [x\ m{=}\ y]$
sledgehammer [*provers = remote-leo2*] **by** *metis*

lemma *eqtest2*: $[x\ m{=}\ y] \Longrightarrow x = y$
sledgehammer [*provers = remote-satallax*] **oops**

Acknowledgments: Nik Sultana, Jasmin Blanchette and Larry Paulson provided very important help on issues related to consistency checking in Isabelle. Jasmin Blanchette instructed us on how to produce latex documents from Isabelle sources and showed us useful tricks in Isabelle.

References

[1] C. Benzmüller and L.C. Paulson. Exploring properties of normal multimodal logics in simple type theory with LEO-II. In *Festschrift in Honor of Peter B. Andrews on His 70th Birthday*, pp. 386–406. College Publications.

[2] C. Benzmüller and L.C. Paulson. Quantified multimodal logics in simple type theory. *Logica Universalis (Special Issue on Multimodal Logics)*, 7(1):7–20, 2013.

[3] C. Benzmüller, F. Theiss, L. Paulson, and A. Fietzke. LEO-II - a cooperative automatic theorem prover for higher-order logic. In *Proc. of IJCAR 2008*, volume 5195 of *LNAI*, pp. 162–170. Springer, 2008.

[4] Y. Bertot and P. Casteran. *Interactive Theorem Proving and Program Development.* Springer, 2004.

[5] J.C. Blanchette, S. Böhme, and L.C. Paulson. Extending Sledgehammer with SMT solvers. *Journal of Automated Reasoning*, 51(1):109–128, 2013.

[6] J.C. Blanchette and T. Nipkow. Nitpick: A counterexample generator for higher-order logic based on a relational model finder. In *Proc. of ITP 2010*, LNCS 6172, pp. 131–146. Springer, 2010.

[7] C.E. Brown. Satallax: An automated higher-order prover. In *Proc. of IJCAR 2012*, LNAI 7364, pp. 111 – 117. Springer, 2012.

[8] K. Gödel. *Appendix A. Notes in Kurt Gödel's Hand*, pp. 144–145. In [12], 2004.

[9] J. Hurd. First-order proof tactics in higher-order logic theorem provers. In *Design and Application of Strategies/Tactics in Higher Order Logics, NASA Tech. Rep. NASA/CP-2003-212448*, 2003.

[10] T. Nipkow, L.C. Paulson, and M. Wenzel. *Isabelle/HOL: A Proof Assistant for Higher-Order Logic.* LNCS 2283. Springer, 2002.

[11] D. Scott. *Appendix B. Notes in Dana Scott's Hand*, pp. 145–146. In [12], 2004.

[12] J.H. Sobel. *Logic and Theism: Arguments for and Against Beliefs in God.* Cambr. U. Press, 2004.

[13] G. Sutcliffe and C. Benzmüller. Automated reasoning in higher-order logic using the TPTP THF infrastructure. *Journal of Formalized Reasoning*, 3(1):1–27, 2010.