CHAPTER XYZ

Analysis of an Ontological Proof Proposed by Leibniz

Matthias Bentert, Christoph Benzmüller, David Streit
and Bruno Woltzenlogel Paleo[1]

One of Leibniz's earliest goals was his ambitious plan, considered to be drafted already in 1668 when he was just 22 years old, to write a collection of Catholic Demonstrations organized in four parts with, respectively, demonstrations of: God's existence; the immortality and incorporeity of the soul; the possibility of the mysteries of the Christian faith; and the authority of the Catholic church and the scripture (Antognazza, 2009)[p. 90].

Although Leibniz pursued this goal throughout his life, and it served as a motivation for him to develop his logic (seen as one of the prolegomena to the demonstrations), he never fully accomplished it. His texts about the topic remained informal, lacking the rigor that would have been possible through his logic.

Today, 300 years after Leibniz's death, in celebration of his lasting legacy to metaphysics, contributions to logic and inspiring foresight of automated reasoning, we accomplish (part of) his goal by showing how one of his informal demonstrations of God's existence could have been formalized in his own Algebra of Concepts. We achieve this through modern automated and interactive theorem provers, and our investigations reveal a few surprises about Leibniz's notions of God and the assumption of its possibility.

---

[1] Author order is alphabetical by surname.

A Brief History of Leibniz's Arguments for God's Existence

Leibniz's first argument for the existence of God was a special case of the cosmological argument resting on the idea that the moving universe requires an incorporeal substance of infinite power (by definition, God) to set it in motion. This argument was presented in his *Dissertation on the Art of Combinations*[2] in 1666 in a very methodical form, with axioms, definitions and a concise step-by-step demonstration. The same argument was presented in an expanded textual form three years later (1669), in his *Confession of Nature against Atheists*.

Between the 18th and the 21st of November 1676, Leibniz visited Spinoza in The Hague (Antognazza, 2009)[p. 177] and discussed, among other topics, ideas from Spinoza's at that time still unpublished *Ethica* (de Spinoza, 1677), which contains an argument for the existence of God, defined as "a substance consisting in infinite attributes, of which each expresses eternal and infinite essentiality". Spinoza's argument is ontological, since it relies on the idea that God's essence involves existence. Soon after the discussion, Leibniz criticized Spinoza's argument in his *Two Notations for Discussion with Spinoza* (November and December 1676), noting gaps in the argument. It is also in these notes that Leibniz famously criticized Descartes's earlier ontological argument (and by extension also Anselm's), where the concept of God is that of "a supremely perfect being" (Ens perfectissum), for being incomplete as it takes for granted that such a concept is possible, without contradiction. He said: "Descartes's reasoning about the existence of a most perfect being assumed that such a being can be conceived or is possible. If it is granted that there is such a concept, it follows at once that this being exists, because we set up this very concept in such a way that it at once contains existence. But it is asked whether it is in our power to set up such a being, or whether such a concept has reality and can be conceived clearly and distinctly, without contradiction. For opponents will say

[2] Leibniz's works cited here can be found in (Leibniz, 1956), (Leibniz, Sämtliche Schriften und Briefe, 1999) or (Leibniz, Sämtliche Schriften und Briefe, 2006).

that such a concept of a most perfect being, or a being which exists through its essence, is a chimera."

Leibniz continued to criticize Spinoza's argument in 1678 (one year after *Ethica*'s publication and Spinoza's death) in his notes *On the Ethics of Benedict de Spinoza* and in 1707 in his *Comments on Spinoza's Philosophy* (Noble, 2010). A major point of contention is the pantheism implied by Spinoza's argument, with Leibniz having stated that: "Among other things, he [Spinoza] believes that the world and God are but a single substantial thing, that God is the substance of all things, and that creatures are only modes or accidents. But I noticed that some of his purported demonstrations, that he showed me, are not exactly right. It is not as easy as one thinks to provide true demonstrations in metaphysics." (Antognazza, 2009)[p.178].

In January 1678, Leibniz sent a *Letter to Henning Huthmann* containing an alternative ontological proof in which God is taken to be an <u>Ens a se, seu Ens ex cujus essentia sequitur existentia, seu Ens necessarium</u> (a self-sufficient being, a being from whose essence its existence follows, a necessary being).

Towards the end of his life, in his *Monadology* (1714), Leibniz presents two arguments for God's existence. The first one can be considered as a more abstract version of his first cosmological argument, relying not on the need for a final cause for the physical universe's movements, but on the need for sufficient reason with a final cause for contingent truths. The second one is the ontological argument with God as an <u>Ens necessarium</u>, completed with the following justification for the possibility of this concept of God: "since nothing can prevent the possibility of that which is without any limits, without any negation, and consequently without any contradiction, this fact alone [i.e. that if God is possible, it necessarily exists] suffices to know the existence of God a priori".

From Metaphysics to Logic

Throughout his life, Leibniz's metaphysical and theological goals seem to have served as a major source of motivation for the development of his logic and mathematics. This can already be seen in his *Dissertation on the Art of Combinations* (1666), which already contains preliminary ideas of his logic and begins with an argument for God's existence. Furthermore, God is mentioned in virtually all of his earlier papers on logic (e.g. *On the General Characteristic* (1679), *On Universal Synthesis and Analysis, or the Art of Discovery and Judgment* (1679), *Two Studies in the Logical Calculus* (1679), *Meditations on Knowledge, Truth and Ideas* (1684)).

In his work *On the Correction of Metaphysics and the Concept of Substance* (1694), he said: "I find that most people who take pleasure in the mathematical sciences shrink away from metaphysics, because they find light in the former but darkness in the latter. […] Yet it seems to me that light and certainty are more needed in metaphysics than in mathematics itself, because mathematical matters carry their own tests and verification with them, this being the strongest reason for success in mathematics. But in metaphysics we lack this advantage entirely. And so a certain distinctive order of procedure is necessary, which, like a thread in a labyrinth, will serve us, no less than the method of Euclid, to analyze our questions in the form of a calculus, yet nonetheless preserving the clarity which should never be lacking from popular speech."

Even in the last years of his life, one of his last works, *The Metaphysical Foundations of Mathematics* (1714), indicates that he had not lost his interest in conciliating the two disciplines.

Leibniz's Algebra of Concepts

Leibniz developed his logical formalism[3] to its most advanced stage in a series of papers from 1686 to 1687 (Leibniz, Sämtliche Schriften und Briefe, 1999). From a modern perspective, the language of Leibniz's logic is a standard first-order language, where terms denote *concepts*. It has two primitive function symbols, denoting *conjunction*[4] and *negation*[5] of concepts, and one primitive binary relation symbol, denoting *containment*[6] of one concept into another. From this small set of primitive functions and relations, others can be defined, such as *subtraction* of concepts and, most interestingly, predicates[7] for *possibility* and *necessity* of concepts. In contrast to the modern modal logic notions of possibility and necessity, which apply to propositions, Leibniz notions apply to concepts. A concept is defined to be possible if it does not contain a contradiction (i.e. a conjunction of a concept and its negation), and necessary if its negation is not possible (cf. *Notiones, Definitiones, Characteres* and *Definitiones: Ens, Possibile, Existens* and *Generales Inquisitiones de Analysis Notionum et Veritatum*). A formalization in Isabelle/HOL of the language of Leibniz's Algebra of Concepts is shown in Figure 1.

---

[3] Our exposition of Leibniz formalism is based on (and agrees with) Lenzen's (Lenzen, Das System der Leibniz'schen Logik, 1990), unless explicitly stated otherwise.

[4] In Leibniz's works, conjunction of two concept terms A and B is usually either denoted by simply concatenating them (i.e. AB) or by using the infix function symbol $\oplus$.

[5] It is important to distinguish conjunction/negation of concepts from conjunction/negation of propositions.

[6] Leibniz often adopts *equality*, depicted as "∞" instead of "=", as the primitive relation symbol, instead of containment. But equality and containment are inter-definable, and we follow Lenzen in choosing containment.

[7] Leibniz actually did not use symbols for the predicates of possibility and necessity, nor for the relation of containment. Such relations were written down in natural language.

```
AoC_Implication.thy (~/Dropbox/Pesquisa/Projects/GoedelGod/Formalizations/Isabelle/Lei...
theory AoC_Implication
imports Main

begin
typedecl c (* Type for concepts *)

consts contains :: "c ⇒ c ⇒ bool" (infix "⊐" 65)
consts conjunction :: "c ⇒ c ⇒ c" (infixr"+" 70)
consts not :: "c ⇒ c" ("~ _" 75)

definition notcontains :: "c ⇒ c ⇒ bool" (infix "⋣" 65) where
          "notcontains A B  ≡ ¬ (A ⊐ B) "
definition equal :: "c ⇒ c ⇒ bool" (infixr "=" 40) where
          "equal A B ≡ A ⊐ B ∧ B ⊐ A"
definition notequal :: "c ⇒ c ⇒ bool" (infixr "≠" 40) where
          "notequal A B ≡  ¬ (A = B)"
(* Note that possible does not mean possible propositions but possible concepts *)
definition possible :: "c ⇒ bool" ("P _" 74) where
          "P B ≡ ∀ A. B ⋣ A + ~ A"
definition necessary :: "c ⇒ bool" ("N _" 74) where
          "N A ≡ ¬ P (~ A) "

definition disjunction :: "c ⇒ c ⇒ c" (infixr "∨" 71) where
          "A ∨ B ≡  ~ ((~A) + ~B)"
(* Note that implication is not introduced by Leibniz or Lenzen *)
definition implication :: "c ⇒ c ⇒ c" (infixr "⟶" 74) where
          "A ⟶ B ≡ ((~ A) ∨ B)"
definition indconcept :: "c ⇒ bool" ("Ind _" 75) where
          "indconcept A  ≡ (P A) ∧ (∀Y. (P (A + Y)) ⟶ A ⊐ Y)"
definition indexists :: "(c  ⇒  bool)  ⇒  bool" (binder "∃" 10) where
          "∃x. A x  ≡ ∃(X::c). (Ind X) ∧ A X"
definition indforall :: "(c  ⇒  bool)  ⇒  bool" (binder "∀" 10) where
          "∀x. A x ≡ ∀(X::c). (Ind X) ⟶ A X"

axiomatization where
 IDEN2: "⋀A B. A = B ⟶ (∀α. α A ⟷ α B)" and
 (* Lenzen uses conjunction. For computational reasons, we use implication. *)
 CONT2: "⋀A B C. A ⊐ B ⟹ B ⊐ C ⟹ A ⊐ C" and
 CONJ1: "⋀A B C. A ⊐ B + C ≡ A ⊐ B ∧ A ⊐ C" and
 NEG1: "⋀A. (~ ~ A) = A" and
 NEG2: "⋀A B. A ⊐ B ≡ (~ B) ⊐ ~ A" and
 (*NEG3 is, contrary to Lenzen's paper, not a theorem. *)
 NEG3: "⋀A. A ≠ ~ A" and
 POSS2: "⋀A B. A ⊐ B ≡ ¬ P(A + ~ B)" and
 (* MAX is equivalent to Lenzen's POSS3. *)
 MAX: "⋀B. P B ⟹ ∃C. ∀A. ((B ⊐ A) ⟶ (C ⊐ A ∧ C ⋣ ~ A))
     ∧ ((B ⊐ ~ A) ⟶(C ⋣ A ∧ C ⊐ ~ A))
     ∧ ((B ⋣ A ∧ B ⋣ ~ A) ⟶ (((C ⊐ ~ A) ∨ C ⊐ A) ∧ (C ⋣ A + ~ A)))"
```

Figure 1: Leibniz's Algebra of Concepts

Isabelle/HOL is an interactive proof assistant based on a higher-order logic. Its expressiveness and user-friendly graphical interface allows the embedding or axiomatization of simpler logical formalisms, such as Leibniz's Algebra of Concepts, in the form of accessible and human-readable higher-order logic theory files.

In addition to the function symbols for conjunction and negation of concepts used by Leibniz, our formalization also declares symbols for *disjunction* and *implication* of concepts, defining them in the usual classical way in terms of the primitive symbols (e.g. disjunction of two concepts is the negation of the conjunction of their negations). Importantly, such defined symbols can be regarded as mere abbreviations for complex expressions and, therefore, do not extend the set of theorems provable in Leibniz's logical formalism.

The consistency of all axioms and definitions shown in Figure 1 can be shown by calling Nitpick, an automated model finder, as seen in Figure 2 below. Nitpick finds a model. Hence, Leibniz's Algebra of Concepts, axiomatized as a higher-order logic theory, is consistent.



Figure 2: Consistency of Leibniz's Algebra of Concepts

From the definitions and axioms shown in Figure 1 above, several useful lemmas can be proven, as listed in Figure 3.

AoC_Implication.thy

AoC_Implication.thy (~/GITHUBS/GoedelGod/Formalizations/Isabelle/Leibniz/)

```
(* CONT1 is not needed as an axiom. *)
lemma CONT1: "A ⊐ A"
  using CONT2 NEG1 equal_def by blast
(* INDEN1 is not needed as an axiom. Lenzen explicitly lists it as one.*)
lemma IDEN1: "A = A"
  by(simp add: CONT1 equal_def)
lemma CONJ2: "A + A = A"
  using CONJ1 CONT1 equal_def by auto
lemma CONJ3: "A + B = B + A"
  using CONJ1 CONT1 equal_def by auto
lemma CONJ4: "A + B ⊐ A"
  using CONJ1 CONT1 by auto
lemma CONJ5: "A + B ⊐ B"
  using CONJ1 CONT1 by auto
lemma NEG4: "A = B ⟹ (A ≠ ~ B)"
  using CONT2 NEG3 equal_def notequal_def IDEN2 by meson
lemma NEG5: "P A ⟹ (A ∉ ~ A)"
  by(simp add: CONJ1 CONT1 possible_def notcontains_def)
lemma NEG6: "P A ⟹ A ⊐ B ⟹ (A ∉ ~ B)"
  by(simp add: CONJ1 possible_def notcontains_def)
(* Lenzen uses conjunction, for computational purposes we use implication *)
lemma POSS1: "A ⊐ B ⟹ P(A) ⟹ P(B)"
  using CONT2 possible_def notcontains_def by blast
lemma NEG7: "(A + ~ A) ⊐ B"
  using CONJ4 CONT1 POSS1 POSS2 by blast
lemma DISJ1: "A ⊐ (A ∨ B)"
  by(metis CONJ3 CONJ4 NEG2 POSS1 POSS2 disjunction_def equal_def)
lemma DISJ2: "B ⊐ (A ∨ B)"
  by(metis CONJ3 CONJ5 NEG2 POSS1 POSS2 disjunction_def equal_def)
(* Lenzen uses conjunction, for computational purposes we use implication *)
lemma DISJ3: "A ⊐ C ⟹ B ⊐ C ⟹ (A ∨ B) ⊐ C"
  by (smt CONJ1 IDEN2 NEG1 NEG2 disjunction_def)
lemma CONT3: "A ⊐ B ≡ (∃Y. A = (B + Y))"
  by (smt CONJ1 CONT1 equal_def)
lemma CONJ6: "∃Y. Y + A ⊐ B"
  using CONJ1 CONT1 by blast
lemma CONJ7: "∃Y. ∃Z. Y + A = Z + B"
  using CONJ3 by blast
lemma NEG8: "A ∉ B ≡ (∃Y. (P(Y  +  A)) ∧ Y + A ⊐ ~ B)"
  by(smt CONJ3 CONJ5 CONT2 NEG6 POSS1 POSS2 equal_def notcontains_def)
lemma CONT4: "A ⊐ B ≡ (∀Y. Y ⊐ A ⟶ Y ⊐ B)"
  by(smt CONT1 CONT2)
lemma CONT5: "A ∉ B ≡ (∀Y. A ≠ Y + B)"
  by(smt CONJ1 CONT1 equal_def notcontains_def notequal_def IDEN2)
lemma IND1: "(Ind A) ≡ (∀Y. (A ⊐ ~ Y) ⟷ A ∉ Y)"
  by (smt CONJ1 CONT4 POSS2 indconcept_def notcontains_def possible_def)
lemma NEG9: "(Ind A) ⟹ A ∉ B ⟹ A ⊐ ~ B"
```

Figure 3: Useful Lemmas of Leibniz's Algebra of Concepts

Leibniz's Argument for the Existence of the Ens Necessarium

Among all of Leibniz's arguments for God's existence, the first ontological argument (of three) in his *Letter to Henning Huthmann* (1678) is the most interesting for a computer-assisted analysis based on Leibniz's own Algebra of Concepts. It is reproduced[8] below:

Theorem: *Si Ens necessarium est possibile, actu existet.*
Proof: *Nam ponamus non existere, inde ratiocinabor hoc modo:*

1) *Ens Necessarium non existit, ex hypothesi.*
2) *Quicquid non existit, illud possibile est non existere.*
3) *Quicquid possibile est non-existere*
   illud falso dicitur non posse non-existere.
4) *Quicquid falso dicitur non posse non existere,*
   illud falso dicitur esse necessarium.
   Nam necessarium est quod non potest non existere.
5) *Ergo Ens necessarium falso dicitur esse necessarium.*
6) *Quae conclusio est vel vera vel falsa.*
7) *Si est vera, sequitur quod Ens necessarium implicet contradictionem, seu sit impossibile, quia de eo demonstrantur contradictoria, scilicet quod non sit necessarium.*
   Conclusio enim contradictoria non nisi de re contradictionem implicante ostendi potest.
8) *Si est falsa, necesse est aliquam ex praemissis esse falsam, sola autem ex praemissis falsa esse potest hypothesis, quod scilicet Ens necessarium non existat.*
9) *Ergo conclusimus*
   Ens necessarium vel esse impossibile, vel existere.
10) *Si ergo Deum definiamus Ens a se, seu Ens ex cujus essentia sequitur existentia, seu Ens necessarium,*
    sequitur Deum si possibilis sit actu esse.

Our translation[9] to English, which is based on Lenzen's translation (Lenzen, Leibniz's Ontological Proof of the Existence of God and

---

[8] The words "Theorem" and "Proof" and the numbering of steps are not in the original. Our numbering is the same as Lenzen's (Lenzen, 2016).

[9] Verb conjugation in Latin is richer than in English. In our translation (as in Lenzen's), Leibniz's use of the subjunctive mood is lost, because we preferred to employ the indicative mood uniformly. For our purposes, this loss is harmless and even elucidative, because neither Leibniz's algebra of

the Problem of "Impossible Objects", 2016) with some modifications[10], is shown below:

Theorem: *If the necessary being is possible, it actually exists.*
Proof: *For if we assume it does not exist, one may reason as follows:*

1) *The necessary being doesn't*[11] *exist, by hypothesis.*
2) *For whatever doesn't exist, for it ~~it~~*[12] *is possible not to exist.*
3) *For whatever ~~it~~ is possible not to exist,*
   of it ~~it~~ is false to say that <u>*it*</u>[13] *cannot*[14] *not exist.*
4) *Of whatever ~~it~~ is false to say that* <u>*it*</u> *cannot not exist,*
   of it ~~it~~ is false to say that <u>*it*</u> *is necessary.*
   For necessary is what cannot not exist.
5) *Therefore, of the necessary being ~~it~~ is false to say* <u>*it*</u> *is necessary.*
6) *This conclusion is either true or false.*
7) *If* <u>*it*</u> *is true, ~~it~~ follows that the necessary being contains a contradiction, i.e. is impossible, because contradictory assertions have been proved about it, namely that it is not necessary. For a*

---

concepts nor any mainstream modern logic has a language capable of expressing mood differences.

[10] The main difference between Lenzen's translation and ours is that Lenzen translates "quicquid" as "whenever something" whereas we translate it as "for/of whatever". Although Lenzen's choice sounds more natural in modern English, we believe "for/of whatever" clearly conveys universal quantification, as intended by Leibniz, whereas the translated sentences with "whenever something" contain donkey pronouns and may suggest existential quantification to readers who are unaware of the pitfalls of donkey anaphora.

[11] The contracted form "doesn't" is chosen as a translation of "non", because "non" is a single word and "does not" would be two words.

[12] When an impersonal Latin verb is translated to modern English, an auxiliary pronoun "it" has to be added. In our translation, all occurrences of such pronouns are stricken through, as "~~it~~".

[13] In contrast to modern English, ellipsis of pronouns is common in Latin. We underline referring pronouns that have been inserted in the translation but omitted through ellipsis in the original.

[14] We translate "non posse" and "non potest" to "cannot", because "posse" and "potest" are conjugated forms of the verb "possum" ("can"). Nevertheless, an alternative translation for step 3, for instance, could be "… to say that ~~it~~ is not possible that <u>it</u> doesn't exist". This alternative would be more similar to the formal language of Leibniz's algebra of concepts, but less similar to his actual original text in Latin.

contradictory conclusion can only be shown about a thing which contains a contradiction.

8) *If it is false, ~~it~~ is needed*[15] *that one of the premises is false. But the only premise that can be false is the hypothesis that the necessary being doesn't exist.*
9) *Hence we conclude that*
   the necessary being either is impossible, or exists.
10) *So if we define God as an "Ens a se", i.e. a being from whose essence existence follows, i.e. a necessary being,*
   ~~it~~ follows that God, if It is possible, actually exists.

This argument is interesting, because it is relatively concise, in comparison to Leibniz's other arguments, and because it uses an informal natural language style and content that seems already quite close to the formal language of his Algebra of Concepts, which was only fully developed 8 to 9 years later. Nevertheless, Leibniz never produced a more rigorous version of the argument above, and thus the question remains: can Leibniz's argument be formalized in his own Algebra of Concepts?

Computer-Assisted Analysis

Our computer-assisted investigation revealed interesting surprises. Figure 4 shows that, if we axiomatize[16] the concept of God as an *Ens necessarium*, i.e. if we state "N(G)" as an axiom, then the argument fails. Nitpick finds a counter-model (of minimum cardinality 4) for the seventh step in Leibniz's argument.

---

[15] "necesse" could also have been translated as "necessary". However, we reserve "necessary" for translations of "necessarium". Translating both as "necessary" would create confusion, especially considering that "necessarium" plays an important role in Leibniz's argument and algebra of concepts, whereas this occurrence of "necesse" is negligible from a logical point of view.

[16] Our axiomatization also states that the concept G is different from E and ~E. These extra axioms are not used in the proof shown in Figure 5. They were added just to prevent Nitpick from generating unnatural counter-models that identified these concepts.



Figure 4: Counter-Model for Proof Attempt with Ens Necessarium

However, if we axiomatize the concept of God as an *Ens ex cujus essentia sequitur existentia*, i.e. if we state “N(G→E)” as an axiom (where “sequitur” is understood as concept implication), the argument goes through. All of Leibniz’s steps are verified by Isabelle/HOL, as shown in Figure 5.

```
theory God_Implication
imports AoC_Implication

begin
consts
  E :: "c" ("E")
  G :: "c" ("G")

definition N :: "c ⇒ bool" where "N A ≡ ¬ P (~ A) "
axiomatization where
 GnotE:     "G ≠ E"      and
 GnotnotE: "G ≠ ~E"      and
 NG:        "N(G ⟶ E)"

(* 2) For whatever doesn't exist, for it is possible not to exist. *)
lemma L2': "(X ∉ E) ⟶ (P (X + ~E))" by (simp add: POSS2 notcontains_def)
(* 3) For whatever it's possible not to exist, of it it's false to say that
it cannot not exist. *)
lemma L3': "(P (X + ~E)) ⟶ ¬¬(P (X + ~E))" by simp
(* 4) Of whatever it is false to say that it is not possible not to exist, of
it's false to say that it is necessary. (For necessary is what cannot not exist.) *)
lemma L4': "¬¬(P (X + ~E)) ⟶ ¬(N (X ⟶ E))" by (smt CONJ1 CONJ4 CONJ5 CONT2 IDEN2
  NEG1 N_def POSS1 disjunction_def equal_def implication_def)
(* 5) Therefore, of the necessary being it's false to say it is necessary. *)
lemma L5': "(G ∉ E) ⟶ ¬(N (G ⟶ E))" using L2' L4' by auto
(* 6) This conclusion is either true or false. *)
lemma L6': "¬(N (G ⟶ E)) ∨ ¬¬(N (G ⟶ E))" by simp
(* 7) If it is true, it follows that the necessary being contains a contradiction, i.e.
is impossible, because contradictory assertions have been proved about it, namely that it
is not necessary. For a contradictory conclusion can only be shown about a thing which
contains a contradiction. *)
lemma L7': "¬(N (G ⟶ E)) ⟶ ¬(P G)" by (simp add: NG)
(* 8) If it is false, necessarily one of the premises must be false. But the only premise
that might be false is the hypothesis that the necessary being doesn't exist. *)
lemma L8': "¬¬(N (G ⟶ E)) ⟶ ¬(G ∉ E)" using L5' by blast
(* 9) Hence we conclude that  the necessary being either is impossible, or exists. *)
lemma L9': "¬(P G) ∨ (G ⊒ E)" using L6' L7' L8' notcontains_def by metis
(* 10)  So if we define God as an "Ens a se", i.e. a being from whose essence its existence
follows, i.e. a necessary being, it follows that God, if It is possible, actually exists. *)
lemma L10': "(P G) ⟶ (G ⊒ E)"  using L9' by auto
(* Note that impossible objects contain any property. Therefore, any impossible object
contains existence *)
lemma God: "(G ⊒ E)" using L5' NG notcontains_def by auto
end
```

Figure 5: Proof for Ens ex Cujus Essentia Sequitur Existentia

Leibniz's argument is verbose. For instance, step 5 of his argument is an instance of the law of excluded middle, and a case distinction on this instance is an unnecessary detour.

Step 10 in Leibniz's proof indicates that Leibniz identified the *Ens necessarium* and the *Ens ex cujus essentia sequitur existentia*. However, with Leibniz's own definitions of necessity, possibility and existence, these two notions of God are distinct.

Furthermore, in the case of the *Ens ex cujus essentia sequitur existentia*, the proviso of possibility (in step 10) is not needed, as shown in Figure 6. This is so, because if the concept of God were impossible, it would easily follow from the definition of possibility that it contain any other concept, including existence. Therefore, Leibniz's criticism that the ontological arguments of Descartes and Anselm are incomplete because they do not establish the possibility of the concept of God does not apply to this version of his ontological argument, even though he apparently did not notice this.

```
God_D_Experiments.thy
God_D_Experiments.thy (~/GITHUBS/GoedelGod/Formalizations/Isabelle/Leibniz/)

theory God_D_Experiments
imports AoC_Implication

begin
consts
 E :: "c" ("E")
 G :: "c" ("G")

definition N :: "c ⇒ bool" where "N A ≡ ¬ P (~ A) "

declare [[ smt_solver = cvc4]]

context (* Short proof with axiom "N(G ⟶ E)" *)
 assumes
  NG: "N(G ⟶ E)"
 begin
  lemma L10: "(P G) ⟶ (G ⊐ E)"
    by (smt CONJ1 CONJ4 CONJ5 CONT2 MAX NEG8 NG N_def POSS1 POSS2 disjunction_def
            implication_def notcontains_def)

  (* This shows that the assumption of possibility is not needed. *)
  lemma L11: "(G ⊐ E)"
    by (metis CONT2 NEG1 NG N_def POSS1 POSS2 disjunction_def equal_def implication_def)

  (* Necessity of concept implication is equivalent to concept containment *)
  lemma L12: "∀X.∀Y.(N(X ⟶ Y) ⟷ (X ⊐ Y))"
    by (metis CONT2 NEG1 N_def POSS1 POSS2 disjunction_def equal_def implication_def)
 end
```

Figure 6: Observations about Possibility, Implication and Containment

Interestingly, in Leibniz's framework, for any two concepts, it is necessary that one implies the other if and only if one contains the other (cf. Figure 6). Therefore, the necessity operator can be regarded as a reflection operator between the type of concepts and the type of propositions.

Other points where Leibniz's informal text lacks precision are his uses of the word "necessarium" ("necessary"). In his later Algebra of Concepts, "necessary" is clearly the dual of "possible". In his ontological argument, however, he says that "necessary is what cannot not exist". That is why occurrences of "it is necessary" in the ontological argument have been formalized as “N(X → E)” instead of “N(X)”. The adequacy of this interpretation of “necessary” and of this formalization is reinforced by the notion of *Ens ex cujus essentia sequitur existentia*, which conveys the intuition of concept implication.

For the observations above to be valuable, it is important to establish that Leibniz’s Algebra of Concepts remains consistent when it is extended with the axiomatization for the *Ens ex cujus essentia sequitur existentia*. For otherwise, anything follows. This can be done with Nitpick, as shown in Figure 7.



Figure 7: Consistency of the Theory where God's Existence is Provable

Although the formal proof shown in Figure 5 verified Leibniz's argument step-by-step, Isabelle/HOL has automated methods that are already powerful enough to prove the final theorem without relying on intermediary lemmas. This can be seen in the proof of Lemma L10 in Figure 6.

Possible Worlds and Modern Modal Logics

Nowadays, words such as "necessity" and "possibility" naturally evoke the modern modal logics having semantics that rely on possible worlds. However, it is crucial to distinguish the current modal logic notions of "necessity" and "possibility" from those of Leibniz's Algebra of Concepts.

From a technical perspective, the Algebra of Concepts talks about necessity and possibility of *concepts*, whereas modal logics talk about necessity and possibility of *propositions*. A proposition is considered possible if it is true in at least one possible world, and necessary if true in all possible worlds.

Furthermore, from a historical perspective, Leibniz was against the idea of possible worlds in December 1676, when he discussed ontological arguments with Spinoza, just one year before he sent to Huthmann the ontological argument reproduced and analyzed here. In his *Two Notations for Discussion with Spinoza*, he wrote[17] that "there is no need of many worlds to increase the multitude of things, for there is no number which is not contained in this one world and, indeed, even in any one of its parts. […] To introduce another kind of existing things, and another world, so to speak, which is also infinite, is to abuse the word 'existence', for we cannot say whether or not these things exist now. […] If all possibles existed, no reason for existence would be needed, and possibility alone would suffice."

---

[17] This comment for Spinoza shows that the common attribution of the idea of possible worlds to Leibniz is not without problems.

Nevertheless, towards the end of his life, in his *Theodicy* (Leibniz, Theodicy, 1710), Leibniz clearly changed his mind and became a firm advocate of possible worlds. What led Leibniz to conclude (and not merely assume) that there must be other possible worlds was the problem of evil. He wrote that "[…] as this vast Region of Verities contains all possibilities, it is necessary that there be an infinitude of possible worlds, that evil enter into various of them, and that even the best of all contain a measure thereof. Thus has God been induced to permit evil" and that "God's decree consists solely in the resolution he forms, after having compared all possible worlds, to choose that one which is the best".

However, even after the *Theodicy*, Leibniz did not seem to have proposed any alternative ontological argument relying on possible worlds and on notions of necessity and possibility of propositions. The argument found in paragraphs 40 to 45 of his *Monadology* (1714), for instance, is still of the same nature as the one reproduced and analyzed here, using necessity and possibility of concepts.

In the 20th Century, with the popularity of possible worlds semantics, there have been several ontological arguments based on modern modal logics. At least two of them are known to have been inspired by Leibniz's ideas: Gödel's ontological argument (Gödel, 1970) and Lenzen's ontological arguments (Lenzen, Das System der Leibniz'schen Logik, 1990) (Lenzen, Leibniz's Ontological Proof of the Existence of God and the Problem of "Impossible Objects", 2016). However, for technical and historical reasons, the use of modern modal logics is probably better attributable to Gödel and Lenzen, and not to Leibniz.

Conclusions

The formalization of Leibniz's ontological argument in his own Algebra of Concepts, as presented here, is historically faithful to the ideals of the young Leibniz at the time when he wrote the argument. The formalization process led to the unexpected discovery that the alternative notions of God as *Ens necessarium* and as *Ens ex cujus essentia sequitur existentia* are actually distinct, according to

Leibniz's own definitions. Leibniz equates both concepts, but his argument succeeds with the former and fails with the latter.

The methodology used in this work, i.e. the use of interactive and automated reasoning tools for metaphysics, has already been used extensively for the analysis of Gödel's ontological argument and its variants (Benzmüller & Woltzenlogel Paleo, 2013-2016), as well as for Anselm's ontological argument (Oppenheimer & Zalta, 2011) (Rushby, 2013). We hope that the use of such reasoning tools will continue to shed light on metaphysics, and that metaphysics, through its modern revival, will once again push the development of logic for the benefit of humankind.

Bibliography

Antognazza, M. R. (2009). *Leibniz: An Intellectual Biography.* New York, NY, USA: Cambridge University Press.

Benzmüller, C., & Woltzenlogel Paleo, B. (2013). Automated Verification and Reconstruction of Gödel's Proof of God's Existence. *Journal of the Austrian Computer Society*, 4-6.

Benzmüller, C., & Woltzenlogel Paleo, B. (2013). Formalization, Mechanization and Automation of Gödel's Proof of God's Existence. *arXiv, 1308.4526*. arXiv.

Benzmüller, C., & Woltzenlogel Paleo, B. (2013). Gödel's God in Isabelle/HOL. *Archive of Formal Proofs*.

Benzmüller, C., & Woltzenlogel Paleo, B. (2013). Gödel's God on the Computer. *10th International Workshop on the Implementation of Logics*, (pp. 1-2).

Benzmüller, C., & Woltzenlogel Paleo, B. (2014). Automating Gödel's Ontological Proof of God's Existence with Higher-Order Theorem Provers. *European Conference on Artificial Intelligence - Frontiers in Artificial Intelligence and Applications. 263.* Prague: IOS Press.

Benzmüller, C., & Woltzenlogel Paleo, B. (2014). On Logic Embeddings and Gödel's God. *22nd International Workshop on Algebraic Development Techniques*, (pp. 8-9).

Benzmüller, C., & Woltzenlogel Paleo, B. (2015). Experiments in Computational Metaphysics. *9th All-India Students' Conference on Science and Spiritual Quest* (pp. 23-40). Kharagpur: Science and Spiritual Quest - Bhaktivedanta Institute.

Benzmüller, C., & Woltzenlogel Paleo, B. (2015). Interacting with Modal Logics in the Coq Proof Assistant. *10th International*

Computer Science Symposium in Russia. 9139, pp. 398-411. Springer.

Benzmüller, C., & Woltzenlogel Paleo, B. (2016). The Inconsistency in Gödel's Ontological Argument: A Success Story for AI in Metaphysics. In S. Kambhampati (Ed.), *International Joint Conference on Artificial Intelligence.* New York: AAAI Press.

Benzmüller, C., & Woltzenlogel Paleo, B. (2016). The Modal Collapse as a Collapse of the Modal Square of Opposition. *Studies in Universal Logic*.

Benzmüller, C., Weber, L., & Woltzenlogel Paleo, B. (2016). Computer-Assisted Analysis of the Anderson-Hájek Ontological Controversy. *Logica Universalis, 10*.

de Spinoza, B. (1677). *Ethica: Ordine Geometrico Demonstrata.*

Gödel, K. (1970). 1970 Manuscript of the Ontological Argument. *Collected Works: Unpublished Essays and Letters, 3*. Oxford University Press.

Leibniz, G. W. (1710). *Essais de Théodicée sur la Bonté de Dieu, la Liberté de l'Homme et l'Origine du Mal* (E-Book ed.). (E. M. Huggard, Trans.) Project Gutenberg.

Leibniz, G. W. (1956). *Philosophical Papers and Letters* (Vol. 1 and 2). (L. E. Loemker, Ed., & L. E. Loemker, Trans.) Chicago, Illinois, USA: The University of Chicago Press.

Leibniz, G. W. (1999). *Sämtliche Schriften und Briefe* (Vol. 4 (6th Series)). (H. Schepers, M. Schneider, G. Biller, U. Franke, & H. Kliege-Biller, Eds.) Germany: Akademie Verlag.

Leibniz, G. W. (2006). *Sämtliche Schriften und Briefe* (Vol. 1 (2nd Series)). (M. Schneider, H. Schepers, P. Beeley, G. Biller, K.-B. Herma, & S. Lorenz, Eds.) Germany: Akademie Verlag.

Lenzen, W. (1990). *Das System der Leibniz'schen Logik.* Berlin: de Gruyter.

Lenzen, W. (2016). Leibniz's Ontological Proof of the Existence of God and the Problem of "Impossible Objects". (R. Silvestre, Ed.) *Logica Universalis*(Special Issue on Logic and Religion).

Noble, C. (2010). Leibniz's Comments on Spinoza's Philosophy. *Fourth Annual Conference of the Leibniz Society of North America*.

Oppenheimer, P., & Zalta, E. (2011). A Computationally-Discovered Simplification of the Ontological Argument. *Australasian Journal of Philosophy, 89*(2), 333-349.

Rushby, J. (2013). The Ontological Argument in PVS. *CAV Workshop "Fun with Formal Methods".* St. Petersburg.

Siders, A., & Woltzenlogel Paleo, B. (Forthcoming). Variants of Gödel's Ontological Proof in a Natural Deduction Calculus.