Gödel's God in Isabelle/HOL

Christoph Benzmüller and Bruno Woltzenlogel Paleo

October 29, 2013

1 Introduction

A formalization and (partial) automation of Dana Scott's version [10] of Goedel's ontological argument [7] in quantified modal logic KB (QML KB) is presented. QML KB is in turn modeled as a fragment of classical higher-order logic (HOL). Thus, the formalization is essentially a formalization in HOL. The employed embedding of QML KB in HOL is adapting the work of Benzmüller and Paulson [2, 1]. Note that the QML KB formalization employs quantification over individuals and quantification over sets of individuals (poperties).

The formalization presented here has been carried and formally verified in the Isabelle/HOL proof assistant; for more information on this system see the textbook by Nipkow, Paulson, and Wenzel [9]. More recent tutorials on Isabelle can be found at the Isabelle homepage: http://isabelle.in.tum.de.

Some further notes:

- 1. This LaTeX text document has been produced automatically from the Isabelle source code document at https://github.com/FormalTheology/GoedelGod/tree/master/Formalizations/Isabelle/GoedelGodSession with the Isabelle build tool.
- 2. The formalization presented here is related to the THF [12] and Coq [4] formalizations available at https://github.com/FormalTheology/GoedelGod/tree/master/Formalizations/.
- 3. All reasoning gaps in Scott's proof script have been automated with Sledgehammer [5] performing remote calls to the higher-order automated theorem prover LEO-II [3]. These calls then suggested respective Metis [8] calls as given below. The Metis proofs are then verified in Isabelle/HOL.
- 4. For consistency checking, the Nitpick model finder [6] has been employed.

2 An Embedding of QML KB in HOL

The types i for possible worlds (or states) and mu for individuals are introduced.

```
\begin{array}{ll} \textbf{typedecl} \ i & -\text{the type for possible worlds} \\ \textbf{typedecl} \ mu & -\text{the type for indiviuals} \end{array}
```

Possible worlds are connected by an accessibility relation.

```
consts r :: i \Rightarrow i \Rightarrow bool (infixr r 70) — accessibility relation r
```

The B axiom (symmetry) for relation r is stated. B is needed only for proving theorem T3.

```
axiomatization where sym: x r y \longrightarrow y r x
```

QML formulas are identified with certain HOL terms of type $i \Rightarrow bool$. This type will be abbreviated in the remainder as σ

```
type-synonym \sigma = (i \Rightarrow bool)
```

The classical connectives \neg, \land, \Rightarrow , and \forall (for individuals and over sets of individuals) and \exists (over individuals) are lifted to type σ . Further connectives could be introduced analogously. *definition* could be used instead of *abbreviation*; the latter are always fully expanded/rewritten, which is fine here, where the focus has been on proof automation, but which would lead to overly complex proof tasks in a purely interactive session.

```
abbreviation mnot :: \sigma \Rightarrow \sigma \ (m\neg) where m\neg \Phi \equiv (\lambda w. \neg \Phi \ w) abbreviation mand :: \sigma \Rightarrow \sigma \Rightarrow \sigma \ (\text{infixr} \ m \land 74) where \Phi \ m \land \ q \equiv (\lambda w. \Phi \ w \land q \ w) abbreviation mimplies :: \sigma \Rightarrow \sigma \Rightarrow \sigma \ (\text{infixr} \ m \Rightarrow 79) where p \ m \Rightarrow \ q \equiv (\lambda w. \ p \ w \longrightarrow q \ w) abbreviation mforall\text{-}ind :: (mu \Rightarrow \sigma) \Rightarrow \sigma \ (\forall i) where \forall i \ P \equiv (\lambda w. \ \forall x. \ P \ x \ w) abbreviation mexists\text{-}ind :: (mu \Rightarrow \sigma) \Rightarrow \sigma \ (\exists i) where \exists i \ P \equiv (\lambda w. \ \exists x. \ P \ x \ w) abbreviation mforall\text{-}indset :: ((mu \Rightarrow \sigma) \Rightarrow \sigma) \Rightarrow \sigma \ (\forall p) where \forall p \ P \equiv (\lambda w. \ \forall x. \ P \ x \ w) abbreviation mbox :: \sigma \Rightarrow \sigma \ (\Box) where \Box \ p \equiv (\lambda w. \ \exists v. \ w \ r \ v \land p \ v) abbreviation mdia :: \sigma \Rightarrow \sigma \ (\lozenge) where \lozenge \ p \equiv (\lambda w. \ \exists v. \ w \ r \ v \land p \ v)
```

For the grounding of lifted formulas the meta-predicate valid is introduced.

```
abbreviation valid :: \sigma \Rightarrow bool([-]) where [p] \equiv \forall w. p w
```

The model finder Nitpick confirms that the axioms and definitions above are consistent. Unfortunately, the respective command syntax for Nitpick is not very intuitive.

```
lemma True nitpick [satisfy, user-axioms, expect = genuine] oops
```

Constant symbol P (Gödel's "Positive") is introduced.

```
consts P :: (mu \Rightarrow \sigma) \Rightarrow \sigma
```

The meaning of P is restricted by axioms A1(a/b): $\forall \phi[P(\neg \phi) \leftrightarrow \neg P(\phi)]$ (Either a property or its negation is positive, but not both.) and A2: $\forall \phi \forall \psi[(P(\phi) \land \Box \forall x[\phi(x) \to \psi(x)]) \to P(\psi)]$ (A property necessarily implied by a positive property is positive.).

axiomatization where

```
A1a: [\forall p \ (\lambda \Phi. \ P \ (\lambda x. \ m \neg \ (\Phi \ x)) \ m \Rightarrow m \neg \ (P \ \Phi))] and A1b: [\forall p \ (\lambda \Phi. \ m \neg \ (P \ \Phi) \ m \Rightarrow P \ (\lambda x. \ m \neg \ (\Phi \ x)))] and A2: [\forall p \ (\lambda \Phi. \ \forall p \ (\lambda \psi. \ (P \ \Phi \ m \land \ \Box \ (\forall i \ (\lambda X. \ \Phi \ X \ m \Rightarrow \psi \ X))) \ m \Rightarrow P \ \psi))]
```

We prove theorem T1: $\forall \varphi[P(\varphi) \to \Diamond \exists x \varphi(x)]$ (Positive properties are possibly exemplified). T1 is proved directly by Sledghammer with command sledgehammer [provers = remote-leo2 remote-satallax]. This successful attempt then suggest to instead try the Metis call in the line below. This Metis call generates a proof object that is verified in Isabelle/HOL's kernel.

```
theorem T1: [\forall p \ (\lambda \Phi. \ P \ \Phi \ m \Rightarrow \Diamond \ (\exists i \ \Phi))] sledgehammer [provers = remote-leo2] using A2 \ A1a by metis
```

Next, the symbol G, for "God-like", is introduced and defined as $G(x) \leftrightarrow \forall \phi[P(\phi) \to \phi(x)]$ (A God-like being possesses all positive properties:).

```
definition G :: mu \Rightarrow \sigma where G = (\lambda x. \forall p \ (\lambda \Phi. P \ \Phi \ m \Rightarrow \Phi \ x))
```

Axiom A3 is added: P(G) (The property of being God-like is positive.). Sledgehammer and Metis then prove corollary $C: \Diamond \exists x G(x)$ (Possibly, God exists.).

```
axiomatization where A3: [P G]
```

```
corollary C: [\lozenge (\exists i \ G)]
sledgehammer [provers = remote-leo2]
using A3 T1 by metis
```

We add axiom $A_4: \forall \phi[P(\phi) \to \Box P(\phi)]$ (Positive properties are necessarily positive).

```
axiomatization where A4: [\forall p \ (\lambda \Phi. \ P \ \Phi \ m \Rightarrow \Box \ (P \ \Phi))]
```

Symbol ess, for "Essence", is introduced and defined as ϕ ess. $x \leftrightarrow \phi(x) \land \forall \psi(\psi(x) \rightarrow \Box \forall y(\phi(y) \rightarrow \psi(y)))$ (An essence of an individual is a property possessed by it and necessarily implying any of its properties.).

```
definition ess :: (mu \Rightarrow \sigma) \Rightarrow mu \Rightarrow \sigma \text{ (infixr } ess 85) \text{ where}

\Phi \ ess \ x = \Phi \ x \ m \land \forall \ p \ (\lambda \psi. \ \psi \ x \ m \Rightarrow \Box \ (\forall \ i \ (\lambda y. \ \Phi \ y \ m \Rightarrow \psi \ y)))
```

Next, Sledgehammer and Metis prove theorem $T2: \forall x [G(x) \to G \text{ ess. } x]$ (Being God-like is an essence of any God-like being).

```
theorem T2: [\forall i \ (\lambda x. \ G \ x \ m \Rightarrow G \ ess \ x)] sledgehammer [provers = remote-leo2] by (metis \ (lifting) \ A1b \ A4 \ G-def \ ess-def)
```

Symbol NE, for "Necessary Existence", is introduced and defined as $NE(x) \leftrightarrow \forall \phi [\phi \ ess. \ x \rightarrow \Box \exists y \phi(y)]$ (Necessary existence of an individual is the necessary exemplification of all its essences.).

```
definition NE :: mu \Rightarrow \sigma where NE = (\lambda x. \ \forall \ p \ (\lambda \Phi. \ \Phi \ ess \ x \ m \Rightarrow \Box \ (\exists \ i \ \Phi)))
```

Moreover, axiom A5 is added: P(NE) (Necessary existence is a positive property.).

```
axiomatization where A5: [P NE]
```

Finally, Sledgehammer and Metis prove the main theorem $T3: \Box \exists x G(x)$ (Necessarily, God exists).

```
theorem T3: [\Box \ (\exists i \ G)]

sledgehammer [provers = remote-leo2]

using A5 \ C \ T2 \ sym \ G-def \ NE-def \ by metis

corollary T4: [\exists i \ G]

sledgehammer [provers = remote-leo2]

using T1 \ T3 \ sym \ G-def \ by metis
```

Finally, the consistency of the entire theory is checked with Nitpick.

```
lemma True nitpick [satisfy, user-axioms, expect = genuine] oops
```

Acknowledgments: Nik Sultana, Jasmin Blanchette and Larry Paulson provided very important help wrt consistency checking in Isabelle. Jasmin Blanchette instructed us on how to produce latex documents from Isabelle sources, and he pointed to many useful tricks in Isabelle.

References

- [1] C. Benzmüller and L.C. Paulson. Exploring properties of normal multimodal logics in simple type theory with LEO-II. In *Festschrift in Honor of Peter B. Andrews on His 70th Birthday*, pp. 386–406. College Publications.
- [2] C. Benzmüller and L.C. Paulson. Quantified multimodal logics in simple type theory. *Logica Universalis (Special Issue on Multimodal Logics)*, 7(1):7–20, 2013.
- [3] C. Benzmüller, F. Theiss, L. Paulson, and A. Fietzke. LEO-II a cooperative automatic theorem prover for higher-order logic. In *Proc. of IJCAR 2008*, volume 5195 of *LNAI*, pp. 162–170. Springer, 2008.
- [4] Y. Bertot and P. Casteran. Interactive Theorem Proving and Program Development. Springer, 2004.
- [5] J.C. Blanchette, S. Böhme, and L.C. Paulson. Extending Sledgehammer with SMT solvers. Journal of Automated Reasoning, 51(1):109–128, 2013.
- [6] J.C. Blanchette and T. Nipkow. Nitpick: A counterexample generator for higher-order logic based on a relational model finder. In *Proc. of ITP 2010*, number 6172 in LNCS, pp. 131–146. Springer, 2010.
- [7] K. Gödel. Appendix A. Notes in Kurt Gödel's Hand, pp. 144–145. In [11], 2004.
- [8] J. Hurd. First-order proof tactics in higher-order logic theorem provers. In *Design and Application of Strategies/Tactics in Higher Order Logics*, NASA Tech. Rep. NASA/CP-2003-212448, 2003.
- [9] T. Nipkow, L.C. Paulson, and M. Wenzel. *Isabelle/HOL: A Proof Assistant for Higher-Order Logic*. Number 2283 in LNCS. Springer, 2002.
- [10] D. Scott. Appendix B. Notes in Dana Scott's Hand, pp. 145-146. In [11], 2004.
- [11] J.H. Sobel. Logic and Theism: Arguments for and Against Beliefs in God. Cambridge U. Press, 2004.
- [12] G. Sutcliffe and C. Benzmüller. Automated reasoning in higher-order logic using the TPTP THF infrastructure. *Journal of Formalized Reasoning*, 3(1):1–27, 2010.